

added_line_options — Options for adding lines to twoway graphs

Description Remarks and examples	Quick start Reference	Syntax Also see	Options
-------------------------------------	--------------------------	--------------------	---------

Description

`yline()`, `xline()`, and `tline()` are used with `twoway` to add lines to the plot region. `tline()` is an extension to `xline()`; see [TS] [tline](#) for examples using `tline()`.

Quick start

Add a horizontal line at the value 0

```
graph_command ..., ... yline(0)
```

Add horizontal lines at the values of 9, 11, and 17

```
graph_command ..., ... yline(9 11 17)
```

Add a vertical line at the value of 17.2

```
graph_command ..., ... xline(17.2)
```

As above, but add a thin red line

```
graph_command ..., ... xline(17.2, lwidth(thin) lcolor(red))
```

As above, but specify line width as 1 point

```
graph_command ..., ... xline(17.2, lwidth(1pt) lcolor(red))
```

Add a vertical line on June 29, 2016

```
graph_command ..., ... tline(29jun2016)
```

Add a red vertical line at 10 and a blue one at 20

```
graph_command ..., ... xline(10, lcolor(red)) xline(20, lcolor(blue))
```

Syntax

<i>added_line_options</i>	Description
<u>y</u> line(<i>linearg</i>)	add horizontal lines at specified <i>y</i> values
<u>x</u> line(<i>linearg</i>)	add vertical lines at specified <i>x</i> values
<u>t</u> line(<i>time_linearg</i>)	add vertical lines at specified <i>t</i> values

yline(), xline(), and tline() are *merged-implicit*; see [G-4] **Concept: repeated options** and see *Interpretation of repeated options* below.

where *linearg* is

```
numlist [ , suboptions ]
```

For a description of *numlist*, see [U] 11.1.8 **numlist**.

and where *time_linearg* is

```
datelist [ , suboptions ]
```

For a description of *datelist*, see [U] 11.1.9 **datelist**.

<i>suboptions</i>	Description
<u>a</u> xis(#)	which axis to use, $1 \leq \# \leq 9$
<u>s</u> tyl(e)(<i>addedlinestyle</i>)	overall style of added line
[<u>n</u> o]e <u>xt</u> end	extend line through plot region's margins
<u>l</u> styl(e)(<i>linestyle</i>)	overall style of line
<u>l</u> patt(ern)(<i>linepatternstyle</i>)	line pattern (solid, dashed, etc.)
<u>l</u> widt(h)(<i>linewidthstyle</i>)	thickness of line
<u>l</u> align(<i>linealignmentstyle</i>)	outline alignment (inside, outside, center)
<u>l</u> col(ore)(<i>colorstyle</i>)	color and opacity of line

Options

yline(*linearg*), xline(*linearg*), and tline(*time_linearg*) specify the *y*, *x*, and *t* (time) values where lines should be added to the plot.

Suboptions

axis(#) is for use only when multiple *y*, *x*, or *t* axes are being used (see [G-3] *axis_choice_options*). axis() specifies to which axis the yline(), xline(), or tline() is to be applied.

styl(e)(*addedlinestyle*) specifies the overall style of the added line, which includes [no]extend and lstyl(e)(*linestyle*) documented below. See [G-4] *addedlinestyle*. The [no]extend and lstyl(e)() options allow you to change the added line's attributes individually, but styl(e)() is the starting point.

You need not specify `style()` just because there is something that you want to change, and in fact, most people seldom specify the `style()` option. You specify `style()` when another style exists that is exactly what you desire or when another style would allow you to specify fewer changes to obtain what you want.

`extend` and `noextend` specify whether the line should extend through the plot region's margin and touch the axis; see [G-3] [region_options](#). Usually `noextend` is the default, and `extend` is the option, but that is determined by the overall `style()` and, of course, the scheme; see [G-4] [Schemes intro](#).

`lstyle(linestyle)`, `lpattern(linepatternstyle)`, `lwidth(linewidthstyle)`, `lalign(linealignmentstyle)`, and `lcolor(colorstyle)` specify the look of the line; see [G-2] [graph twoway line](#). `lstyle()` can be of particular use:

To create a line with the same look as the lines used to draw axes, specify `lstyle(foreground)`.

To create a line with the same look as the lines used to draw grid lines, specify `lstyle(grid)`.

Remarks and examples

[stata.com](http://www.stata.com)

`ylines()` and `xlines()` add lines where specified. If, however, your interest is in obtaining grid lines, see the `grid` option in [G-3] [axis_label_options](#).

Remarks are presented under the following headings:

Typical use
Interpretation of repeated options

Typical use

`ylines()` or `xlines()` are typically used to add reference values:

```
. scatter yvar xvar, yline(10)
. scatter yvar year, xline(1944 1989)
```

To give the line in the first example the same look as used to draw an axis, we could specify

```
. scatter yvar xvar, yline(10, lstyle(foreground))
```

If we wanted to give the lines used in the second example the same look as used to draw grids, we could specify

```
. scatter yvar year, xline(1944 1989, lstyle(grid))
```

Interpretation of repeated options

Options `ylines()` and `xlines()` may be repeated, and each is executed separately. Thus different styles can be used for different lines on the same graph:

```
. scatter yvar year, xline(1944) xline(1989, lwidth(3))
```

Reference

Cox, N. J. 2009. [Stata tip 82: Grounds for grids on graphs](#). *Stata Journal* 9: 648–651.

Also see

[G-4] *addedlinestyle* — Choices for overall look of added lines

[G-4] *colorstyle* — Choices for color

[G-4] *linealignmentstyle* — Choices for whether outlines are inside, outside, or centered

[G-4] *linepatternstyle* — Choices for whether lines are solid, dashed, etc.

[G-4] *linestyle* — Choices for overall look of lines

[G-4] *linewidthstyle* — Choices for thickness of lines